# Maintainer: Maxime Gauduin <alucryd@archlinux.org>
# Contributor: Steven Allen <steven@stebalien.com>
# Contributor: Limao Luo <luolimao+AUR@gmail.com>
# Contributor: Wieland Hoffmann <the_mineo@web.de>
# Contributor: Amr Hassan <amr.hassan@gmail.com>

pkgname=python-pylast
pkgver=5.1.0
pkgrel=2
pkgdesc='A Python interface to Last.fm and Libre.fm'
arch=(any)
url='https://github.com/pylast/pylast'
license=(Apache)
depends=(
  python-httpx
  python-importlib-metadata
)
makedepends=(
  git
  python-build
  python-hatch-vcs
  python-hatchling
  python-installer
)
_tag=ce76c03581511499cf3d2163e0465c9aa05141a9
source=(git+https://github.com/pylast/pylast.git#tag=${_tag})
sha256sums=(SKIP)

pkgver() {
  cd pylast
  git describe --tags
}

build() {
  cd pylast
  python -m build --wheel --no-isolation
}

package() {
  python -m installer --destdir="${pkgdir}" pylast/dist/*.whl
}

# vim: ts=2 sw=2 et:
